# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-cgi-session
pkgver=4.48
pkgrel=10
pkgdesc="Perl/CPAN Module CGI::Session : Session management in CGI  applications"
arch=(any)
url="https://search.cpan.org/dist/CGI-Session/"
license=("GPL" "PerlArtistic")
source=("https://www.cpan.org/authors/id/M/MA/MARKSTOS/CGI-Session-$pkgver.tar.gz")
sha256sums=('46756461c24ce7666b8108ddb96dbab612699df3012c80ef11016619fe1554f7')

build() {
  cd "$srcdir"/CGI-Session-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "$srcdir"/CGI-Session-$pkgver
  make pure_install doc_install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
